Розв’язки задач першого туру 

обласної Інтернет-олімпіади з програмування 

1. Задача  PARK
(20 балів)






  Код VIO_1_1

Ім’я вхідного файла:   PARK.DAT
Ім’я вихідного файла: PARK.SOL

Максимальний час роботи на одному тесті: 5с

У парку розваг дирекція вирішила прикрасити центральну алею. Було запропоновано пофарбувати доріжку алеї різними кольрами. Для цього придбали N банок фарби різного кольору (1<=N<=60000). До фарбування алеї було залучено N учнів сусідньої школи. Оскільки за учнями ніхто не слідкував, то вони зафарбовували алею в тих місцях, які їм сподобались і навіть фарбували ті ділянки алеї, які вже були зафарбовані іншими учнями. 

Допоможіть дирекції парку:

· обчислити довжину зафарбованої ділянки алеї;

· кількість ділянок, яку зафарбовував лише один учень.  

Формат вхідних даних:

Перший рядок вхідного файлу park.dat містить ціле число d (d>=60000) – довжину алеї, яку мали зафарбувати. Другий рядок містить ціле число N (1<=N<=60000) – кількість учнів, які фарбували алею. Наступні N рядків містять пари двох цілих чисел аі і ві (0<=ai<d, 0<bi<=d), що розділені одним пропуском, які є координатами відповідно початку і кінця замальованої ділянки алеї парку одним учнем.

Формат вихідних даних:

У першому рядку вихідного файлу park.sol вивести довжину зафарбованої ділянки алеї. У другому рядку вивести кількість ділянок, які зафарбовував лише один учень.  

Приклад вхідних і вихідних даних:

Існує декілька підходів до розв'язання цієї задачі. Вам пропонується варіант розв’язку цієї задачі в середовищі програмування Turbo Pascal. 
Обчислити довжину зафарбованої ділянки алеї досить просто. Зчитуєм координати відрізка АіВі (Аі(х1), Ві(х2)). В масиві а[1..60000] нульові елементи а[х1+1], а[х1+2], …, а[х2] замінюємо на 1 та збільшуємо довжину зафарбованої ділянки на 1, а ненульові (які були раніше зафарбовані) замінюємо на 2 не змінюючи довжину зафарбованої ділянки.

Ділянку замальовував лише один учень, якщо кількість одиничних елементів а[х1+1], а[х1+2], …, а[х2] рівна  довжині відрізка АіВі. Якщо дана умова виконується лічильник кількості шуканих ділянок збільшуємо на 1. Перевіряти дану умову необхідно коли переглянули всі відрізки (наступні учні можуть пошкодити попередні ділянки). В середовищі програмування Turbo Pascal неможливо зберігати одночасно і масив а[1..60000] і координати всіх відрізків, тому вхідні дані необхідно ще раз зчитувати з вхідного файлу. 
В мовах програмування, які дозволяють створювати великі масиви повторного зчитування вхідних даних робити недоцільно. Для прискорення часу виконання програми можна запам’ятовувати лише координати тих відрізків, в яких всі елементи  а[х1+1], а[х1+2], …, а[х2] рівні 1 при заповнені масиву а[1..60000].

Мова програмування: Pascal 

Середовище програмування: Turbo Pascal 7.0

Program Parc;

var a:array[1..60000] of byte;

    k,n,d,i,j,x1,x2,s,s2:longint;

    f_in,f_out:text;

begin

  assign(f_in,'park.dat');

  assign(f_out,'park.sol');

  reset(f_in);

  readln(f_in,d);

  readln(f_in,n);

  s:=0;

  for i:=1 to n do

    begin

      readln(f_in,x1,x2);

      for j:=x1+1 to x2 do

        begin

          if a[j]=0 then

            begin

              a[j]:=1;

              s:=s+1

            end else a[j]:=2;

        end;

    end;

  close(f_in);

  rewrite(f_out);

  writeln(f_out,s);

  reset(f_in);

  readln(f_in);

  readln(f_in);

  k:=0;

  for i:=1 to n do

    begin

      readln(f_in,x1,x2);

      j:=x1+1;

      s2:=0;

      while (a[j]=1) and (j<=x2) do

        begin

          inc(s2);

          inc(j);

        end;

      if s2=x2-x1 then k:=k+1;

    end;

  writeln(f_out,k);

  close(f_in);

  close(f_out);

end.
2. Задача  сyclone
(100 балів)


     
                       Код VIO_1_2

Ім’я вхідного файла:   сyclone.dat 

Ім’я вихідного файла: сyclone.sol
Максимальний час роботи на одному тесті: 5 с.

Давним-давно в одній далекій-далекій галактиці була собі планета Бурабумбія, яка вирізнялася своїми погодними умовами. Вся планета була дуже різноманітною за природними умовами, а у її небі і над землею постійно вирували сильні циклони. Космологи планети Земля, які вперше відкрили нову планету,  для дослідження умов життя на ній відправили на її поверхню зонд, який безперервно пересилав отримані дані про флору і фауну на космічний корабель. Проте часті циклони, які виникали в атмосфері і проходили прямо над зондом, створювали перешкоди у передачі інформації і сигнал автоматично переривався, що дуже сердило вчених. Знічев’я вони підрахували, що в небі над Бурабумбією знаходиться 

 циклонів і всі вони рухаються в одному напрямку і з постійною швидкістю 

 точка циклону з початковими координатами 

Заради спрощення вчені вважають, що кожний циклон є многокутником (враховуються його сторони), вершини якого мають цілі координати. Цей многокутник не обов’язково є опуклим, але він без самоперетинів. Циклони можуть поєднуватися, тобто многокутники, що їх відображають, можуть перетинатися.

На землі знаходиться контрольний зонд у точці 

, а прямо над ним, вище атмосфери і всіх циклонів, завис космічний корабель, який сприймає сигнал (сигнал передається прямолінійно). І ось, нудьгуючи без роботи, вчені зацікавилися, скільки разів сигнал перериватиметься через погодні умови. Навіть, якщо тільки одна точка циклону перекриває сигнал, то сигнал переривається. Вчені знають, що у початковий момент часу жоден циклон не нависав над зондом. Ваше завдання: написати програму, яка дає відповідь на запитання вчених. 

Формат вхідних даних:

Перший рядок вхідного файлу cyclone.dat містить три цілих числа 

, які розділені пропуском, 

 - кількість циклонів у небі над Бурабумбією, а 

 вектор швидкості переміщення циклонів. Координата 

 визначає західно-східне переміщення циклонів, а 

 північно-південний напрямок переміщення.

Наступні 

 рядків містять опис циклонів (по одному на рядок). Перше число у кожному рядку - 

- кількість вершин циклону, 

 цілих чисел: 

 - координати вершин многокутника 

 у порядку його обходу за годинниковою стрілкою. Всі числа розділені одним пропуском. 

Сигнал може перериватися не більше, ніж 

 раз.

Формат вихідних даних:

У єдиному рядку вихідного файлу cyclone.sol повинно міститися єдине ціле число - кількість разів переривання сигналу між зондом і космічним кораблем.

Приклад вхідних і вихідних даних:

	сyclone.dat
	сyclone.sol

	4  -2  -1
4 6 2 6 4 8  4 8 2
4 2 3 1 -1 2 5 4 2
3 -3 1 -1 2 -1  -1
5 5 3 3 3 3 5 6 5 6 -1

Малюнок до вхідного файлу:

Щоб розв’язати дану задачу необхідно знайти всі точки перетину вектора ОА (вектор ОА має напрям протилежний до напряму руху циклонів) з сторонами многокутників. Не всі точки перетину є початком або кінцем переривання. Перше переривання здійснить точка А1 і закінтиться воно в точці А2.  Друге переривання здійснить точка А3 (миттєве переривання). Третє перериванння почнеться в точці А4 і завершиться в точці А5. В точці В1 триває третє переривання. Отже, вся задача зводиться до того, щоб визначити які точки є початком переривання, які кінцем, а які лежать всередині переривання.

Пряму, що проходить через точки 

 назвемо прямою вектора швидкості. 

Читаємо з файлу кількість циклонів, і вектор швидкості переміщення циклонів. Тоді читаємо розмір многокутника і починаємо працювати з ним. Зберігаємо у масиві cloud координати точок даного многокутника. Введемо додаткову функцію Side. Вона може набувати трьох значень і визначає чи яким чином проходить якась дана точка многокутника відносно контрольного зонду. Якщо сама точка проходить нижче зонду, то значення Side буде 1; якщо вище від зонду, то -1, якщо проходить саме по зонду, то 0. Спочатку шукаємо першу таку точку многокутника, що вона не проходить по зонду (тобто не лежить на прямій вектора швидкості). Далі, знайшовши першу таку точку виконуємо наступні дії: розглядаємо наступну за «обходом» точку многокутника  Застосовуємо до неї функцію Side. Тепер, якщо добуток значень функцій для обох точок дорівнює 1, то це означає, що обидві вони лежать або нижче, або вище від вектора швидкості, тобто ребро, що їх сполучає не перетинає пряму вектора швидкості, тобто не існує перериву сигналу з цією стороною многокутника і зондом. 

function side(a : point) : shortint;

VAR

    x, wx, wy : int64;

        kx, ky : int64;

BEGIN

  wx := a.x;

  wy := a.y;

  kx := vel_x;

  ky := vel_y;

  x := kx * wy - ky * wx;{косий добуток для точки}

  if x = 0 then side := 0 else

  if x > 0  then side := 1 else

  side := -1;

END;

Якщо ж добуток значень функцій дорівнює -1, то це означає, що точки многокутника лежать по різні сторони від прямої вектора швидкості і ми продовжуємо перевіряти далі. 

Тобто, ми будемо заносити для розгляду ті ребра, які є підозрілими на переривання. Зрозуміло, що перше ребро, яке перетинає пряма вектора швидкості буде ребром переривання точно.

Тепер помітимо, що для розглядання ребра підозрілого на переривання введемо допоміжну процедуру, яка шукатиме точку перетину  ребра заданого двома даними точками і прямої вектора швидкості. Зважаємо на те, що пряма вектора швидкості може перетинати ребра трьома способами:

· пряма перетинає ребро висхідного напрямку:  (наприклад ребро K2K3 і точка A1 на ньому);
· пряма перетинає ребро нисхідного напрямку : (наприклад ребро K1K2 і точка A2 на ньому);
· пряма перетинає ребро по точці: (наприклад точка K4=A3)
Тепер, згідно цього відмічаємо ребро як проглянуте і запам’ятовуємо його (точку перетину) та характер перетину прямої вектора швидкості цього ребра. 

Якщо ж добутки є додатніми, то не змінюємо характер положення ребра, також заносячи його у масив.

procedure Add_intersection(a, b : point; cloud : integer; kind : byte);

VAR c : crossing;

    wx, wy : int64;

BEGIN

  wx := a.x;

  wy := a.y;

  c.posh := b.y * wx - b.x * wy;

  c.posl := vel_y * (wx - b.x) - vel_x * (wy - b.y);

  if (c.posh < 0) AND (c.posl < 0) then

  begin

    c.posh := -c.posh;

  c.posl := -c.posl;

  end else

  if (c.posh < 0) OR (c.posl < 0) then exit;

  c.cloud := cloud;

  c.kind := kind;

  cr[cr_nr] := c;

  inc(cr_nr);

END;

Якщо ж добуток значень функції дорівнює нулю, то це означає, що друга взята нами точка належить прямій вектора швидкості, а тому ми продовжуємо перевірку записуючи у масив наступне ребро.

Тоді, ми сортуємо отримані точки ребер і рахуємо кількість переривань таким чином:

· якщо спочатку перетинає ребро висхідного напряму, а потім нисхідного, то зараховуємо переривання, як таке, що відбулося;

· якщо спочатку перетинає «висхідне» ребро, а потім знов висхідне, або точку, то переривання триває (не збільшуємо кількість переривань, а для точки додатково перівіряємо переривання);

· якщо на деякому кроці висхідне ребро і нисхідне співпали (саме в такому порядку), то переривання триває, інакше переривання закінчилося;

· випадок, коли пряма перетинає точку включається у перший пункт (по перетин точки спільної для ребер буде різного характеру)

Проглянувши таким чином всі ребра, що нас цікавлять матимемо остаточний результат кількості переривань сигналу, які визначаємо згідно вищезгаданих правил.

function Count_Result : longint;

VAR inside  : ARRAY[0..MAX_CLOUDS - 1] of boolean;  
  on_edge : ARRAY[0..MAX_CLOUDS - 1] of boolean; 

  am_edges, am_inside, result, x : longint;

  location : crossing;

BEGIN

  am_inside := 0;

  am_edges := 0;

  result := 0;

  location.posh := 0;  
  location.posl := 1; 

  for x := 0 to clouds - 1 do

  begin

    inside[x] := false;

    on_edge[x] := false;

  end;

  for x := 0 to cr_nr - 1 do

  begin

    if (am_inside = 0) AND (am_edges = 0) AND (compare(location, cr[x]) <> 0) then

      inc(result);

    location := cr[x];

    if (location.kind = 1) OR (location.kind = 2) then 

    begin

      on_edge[location.cloud]:=NOT on_edge[location.cloud];

      if on_edge[location.cloud] then inc(am_edges) else dec(am_edges);

    end;

    if (location.kind = 0) OR (location.kind = 2) then 

    begin

      inside[location.cloud] := NOT inside[location.cloud];

      if inside[location.cloud] then inc(am_inside) else dec(am_inside);

    end;

  end;

  Count_Result := result;

END;

Остаточно код правильної програми буде таким скомпільованої в середовище програмування: Free Pascal 2.2.2

{$R-}

const MAX_CLOUDS = 1000;

const CLOUD_SIZE = 1000;

var high_part, low_part  : qword;

TYPE point = RECORD

    x, y : longint;

END;

TYPE superlong = RECORD

    high, low : qword;

END;

TYPE crossing = RECORD

  posh, posl : int64;  { Відносна позиція точки }

  cloud : integer;     { Власник точки}

  kind  : byte;        { Тип перетину}

END;

VAR

    clouds : integer;

    vel_x, vel_y : longint; { Вектор швидкості }

    cr_nr  : longint;

    cr : ARRAY [0..MAX_CLOUDS * CLOUD_SIZE] of crossing; {точки перетину}

function multiply(v1, v2 : qword) : superlong;

{Перемножує два значення типу qword і повертає результат як superlong

У цій функції приймається, що величини які множаться є щонайбільше 62-bit довжини }

VAR    x : qword;

BEGIN

  x := (v1 AND low_part) * (v2 shr 31) + (v1 shr 31) * (v2 AND low_part);

  multiply.low := (v1 AND low_part) * (v2 AND low_part);

  multiply.high := (v1 shr 31) * (v2 shr 31);

  multiply.low  := multiply.low + ((x AND low_part) shl 31);

  multiply.high := multiply.high + (x shr 31) + (multiply.low shr 62);

  multiply.low  := multiply.low AND high_part;

END;

function compare(a, b : crossing) : shortint;

{Порівнює позицію двох точок}

VAR

    a1, b1 : superlong;

BEGIN

  a1 := multiply(a.posh, b.posl);

  b1 := multiply(a.posl, b.posh);

  if a1.high = b1.high then

  begin

    if a1.low = b1.low then compare := 0 else

    if a1.low < b1.low then compare := -1 else

    compare := 1;

  end else

  if a1.high < b1.high then compare := -1 else

  compare := 1;

END;

function side(a : point) : shortint;

{Визначає позицію заданої точки відносно вектору швидкості VAR

    x, wx, wy : int64;

    kx, ky : int64;

BEGIN

  wx := a.x;

  wy := a.y;

  kx := vel_x;

  ky := vel_y;

  x := kx * wy - ky * wx;

  if x = 0 then side := 0 else

  if x > 0  then side := 1 else

  side := -1;

END;

procedure Add_intersection(a, b : point; cloud : integer; kind : byte);

{Перевіряє наявність перетин між ребром і вектором швидкості} 

VAR

    c : crossing;

    wx, wy : int64;

BEGIN

  wx := a.x;

  wy := a.y;

  c.posh := b.y * wx - b.x * wy;

  c.posl := vel_y * (wx - b.x) - vel_x * (wy - b.y);

  if (c.posh < 0) AND (c.posl < 0) then

  begin

    c.posh := -c.posh;

    c.posl := -c.posl;

  end else

  if (c.posh < 0) OR (c.posl < 0) then exit;

  c.cloud := cloud;

  c.kind := kind;

  cr[cr_nr] := c;

  inc(cr_nr);

END;

procedure Read_Data (fName: String);

VAR

    kind : byte;

    size, x, y, beg : integer;

    pos, f_side, l_side : integer;

    cloud  : ARRAY [0..CLOUD_SIZE-1] of point;

    F: Text;

BEGIN

  Assign(F, fName);

  Reset(F);

  Read(F, clouds);

  Read(F, vel_x);

  Read(F, vel_y);

  vel_x := -vel_x;

  vel_y := -vel_y;

  for x := 0 to clouds - 1 do

  begin

    Write('.');

    Read(F, size);

    for y := 0 to size - 1 do 

      begin

       Read(F, cloud[y].x);

       Read(F, cloud[y].y);

      end;

    pos := 0;

    while side(cloud[pos]) = 0 do

    begin

     inc(pos); {Знаходить вершину, що не лежить на вектоі швидкості}

    end;

    f_side := side(cloud[pos]);

    y := 1;

    while y <= size do

    begin

      l_side := f_side;

      f_side := side(cloud[(pos + y) MOD size]);

      if l_side * f_side = 1 then

        inc(y); { Вершини знаходяться по один бік}

      if l_side * f_side = -1 then

      begin { Вершини знаходяться по різні сторони від прямої вектора швидкості }

        Add_intersection(cloud[(size + pos + y - 1) MOD size], cloud[(pos + y) MOD size], x, 0);

        inc(y);

      end;

      if l_side * f_side = 0 then

      begin { Подальша перевірка }

        beg := pos + y;

        while side(cloud[(pos + y) MOD size]) = 0 do 

        begin 

          inc(y); 

        end;

        f_side := side(cloud[(pos + y) MOD size]);

        if pos + y <> beg + 1 then

        begin

          if l_side = f_side then kind := 1 else kind := 2;

          Add_intersection(cloud[(pos + y) MOD size], cloud[(pos + y - 1) MOD size], x, kind);

          Add_intersection(cloud[(size + beg - 1) MOD size], cloud[beg MOD size], x, 1);

        end else

        begin

          if l_side = f_side then kind := 3 else kind := 0;

          Add_intersection(cloud[(pos + y) MOD size], cloud[(size + pos + y - 1) MOD size], x, kind);

        end;

      end;

    end;

  end;

  Close(F);

END;

function Count_Result : longint;

{Шукає відсортований список точок перетину і підраховує результат} VAR

  inside  : ARRAY[0..MAX_CLOUDS - 1] of boolean;

  on_edge : ARRAY[0..MAX_CLOUDS - 1] of boolean; 

  am_edges, am_inside, result, x : longint;

  location : crossing;

BEGIN

  am_inside := 0;

  am_edges := 0;

  result := 0;

  location.posh := 0;

  location.posl := 1; 

  for x := 0 to clouds - 1 do

  begin

    inside[x] := false;

    on_edge[x] := false;

  end;

  for x := 0 to cr_nr - 1 do

  begin

    if (am_inside = 0) AND (am_edges = 0) AND (compare(location, cr[x]) <> 0) then

      inc(result);

    location := cr[x];

    if (location.kind = 1) OR (location.kind = 2) then {Перетин міняє статус ребра}

    begin

      on_edge[location.cloud]:=NOT on_edge[location.cloud];

      if on_edge[location.cloud] then inc(am_edges) else dec(am_edges);

    end;

    if (location.kind = 0) OR (location.kind = 2) then {Перетин міняє статус хмари}

    begin

      inside[location.cloud] := NOT inside[location.cloud];

      if inside[location.cloud] then inc(am_inside) else dec(am_inside);

    end;

  end;

  Count_Result := result;

END;

procedure Rotate(b, e : longint);

VAR

  tr       : crossing;

  x, swap : longint;

BEGIN

  for x := b to e do

  begin

    swap := random(e - b + 1) + b;

    tr := cr[x];

    cr[x] := cr[swap];

    cr[swap] := tr;

  end;

END;

function partition(b, e : longint) : longint;

VAR

  x, tr : crossing;

BEGIN

  x := cr[b];

  dec(b);

  inc(e);

  while true do

  begin

    repeat dec(e); until compare(x, cr[e]) <> -1;

    repeat inc(b); until compare(x, cr[b]) <> 1;

    if b < e then

    begin

      tr := cr[e];

      cr[e] := cr[b];

      cr[b] := tr;

    end else

    begin

      partition := e;

      exit;

    end;

  end;

END;

procedure quicksort(b, e : longint);

VAR

  el : longint;

BEGIN

  if b < e then

  begin

    el := partition(b, e);

    quicksort(b, el);

    quicksort(el + 1, e);

  end;

END;

var i,j:longint;


tmp:crossing;

  Res: longint;

  FNameIn, FNameOut: String;

  F: Text;

BEGIN

  FNameIn:=ParamStr (1);

  FNameOut:=ConCat(Copy(FNameIn, 1, Pos('.', FNameIn)-1), '.out');

  low_part := 2147483647;                 { 2^31 - 1 }

  high_part := low_part * (low_part + 2);

 { 2^62 - 1 }

  Randomize();

  cr_nr := 0;

  Read_Data(FNameIn);

  WriteLn('.');

  Rotate(0, cr_nr - 1);

  quicksort(0, cr_nr - 1);

  Res:= Count_Result();

  Assign(F, FNameOut);

  Rewrite(F);

  Write(F, Res);

  Close(F);

  WriteLn('****Done****');

END.

Приклад виконання :

Вхідний файл  Cyclony0.dat

Командний рядок :

clo0.exe Cyclony0.dat

Вихідний файл буде Cyclony0.out

